
1 – Présentation 
 

 KoXo Charter est un outil qui permet d’afficher en plein écran une charte informatique personnalisée lors de 
la première ouverture de session dans un réseau basé sur Active Directory. KoXo Charter ne fonctionne pas dans un 
domaine de type NT4 ou émulation comme Samba, étant donné qu’il nécessite un vrai annuaire. 
 KoXo Charter permet d’afficher une page Web qui est personnalisée qui pourra comporter le nom de 
l’utilisateur, le nom de l’ordinateur, le nom de l’établissement etc.  
 Lorsque l’utilisateur se logue il pourra accepter ou refuser la charte, un fichier de compte rendu sera généré. 
Selon la politique de l’établissement on pourra décider de clôturer la session si l’utilisateur n’accepte pas la charte. 
 L’administrateur pourra configurer son propre modèle de charte avec les variables qui permettront 
d’afficher des informations de l’utilisateur, de la machine, du serveur, la date et l’heure. 
 

Exemple d’écran obtenu :  

2 – Installation 
 

KoXo Charter est fourni avec les fichiers et dossiers suivants : 

• KoXoCharter.exe 

• KoXoCharter.vbs 

• KoXoCharterGPO.vbs 
Exemple de charte :  

• Charte.docx 

• Charte.htm 

• Charte_fichiers 
 

 En réalité seul l’exécutable est nécessaire s’il est appelé avec les bon paramètres. Le fichier KoXoCharter.vbs 
permet d’appeler l’exécutable de manière plus simple pour l’administrateur. 
 
 Pour installer KoXo Charter dans votre réseau il suffit de copier le dossier complet KoXoCharter dans un 
partage accessible à tous : exemple le partage NETLOGON. 
 
Pour que KoXo Charter soit appelé il y a plusieurs méthodes : 
 

1. Par le script de login (VBS ou BAT, seule la démarche en VBS est détaillée ici car les fichiers BATCH ne sont 
plus d’actualité). 

2. Par les stratégies de groupes. 
 
2-1 Lancement par script de login (vbs) 
 

 Dans les premières versions de KoXo Administrator, au niveau des groupes primaires concernés, il faut 
ajouter en fin de script par exemple (ou même au début) la/les  ligne(s) suivante(s) : 
 
Set Wscr=CreateObject("Wscript.Shell")  (Si necessaire) 

Wscr.Run "\\%SERVER[1]%\NETLOGON\KoXoCharter\KoXoCharter.VBS", 1, True 

 
Dans les scripts des dernières versions de KoXo Administrator il suffit de rajouter :  
 
Execute GetLogonServer & "\NETLOGON\KoXoCharter\KoXoCharter.VBS" 

 
N.B. : Cette dernière solution est la meilleure car il y a une gestion des erreurs et une optimisation de la bande 
passante pour les installations multi sites. 
 
Il suffit ensuite de propager ce script aux groupes secondaires. 

2-1 Lancement par stratégies de groupes 
 
Il faut tout d’abord choisir ou créer une GPO qui concerne les utilisateurs visés, ensuite se positionner au niveau des 
scripts de login et créer un script dont le nom est CharterGPO.vbs. 
 

 

Ajout du nom du script : 
 

 
 
 Cliquer ensuite sur « Afficher les fichiers » et copier/coller le fichier « CharterGPO.vbs » fourni. 
 
Contenu du fichier « KoXoCharterGPO.vbs » :  
 
 

'///////////////////////////////////////////////////////////////////// 

'//             (C)2006-2019 KoXo Développement                     // 
'//            KoXo Charter V2.0 : Script for GPO                   // 
'///////////////////////////////////////////////////////////////////// 

Dim Wscr 
Set Wscr=CreateObject("Wscript.Shell") 
S=Mid(WScript.ScriptFullName,3,255) 

S=Mid(S,1,InStr(S,"\")) 

S="\\"+S+"NETLOGON\KoXoCharter\KoXoCharter.VBS" 

Wscr.Run S, 1, True 
WScript.Quit 

3 – Elaboration d’une charte 
 

Création du fichier sous Microsoft Word : 
 

 
 

 Il suffit donc de saisir le document voulu avec des variables réservées (voir le chapitre « Annexes »), on peut 
aussi placer des boutons (images) sur lesquelles on place des liens spéciaux du type :  
 

• koxocharter://accept/   Acceptation de la charte 

• koxocharter://refuse/   Refus de la charte 

• koxocharter://refuse:logoff  Refus de la charte et fermeture de session 

Pagination :  
 
Afin d'avoir un bon affichage de la page Web, il penser à ce que la taille n'est pas une page A4 mais un écran dont 
l'aspect n'est aps le même. Pour afficher proprement les boutons il est conseillé d'utiliser un tableau et de raisonner 
pour une page Web (utiliser des tableaux). 
 

 
Images dans un tableau en bas de page 

 
Une fois le fichier créé il faudra le sauvegarder sous le format « Page Web filtrée ». 
 
4 – Paramètres de lancement 
 

 L’exécutable « KoXoCharter.exe »  est fourni avec un fichier VBS qui l’appelle, les options de ligne de 
commande sont les suivantes : 
 

KoXoCharter.exe    FichierHTM      FichierReponse     [/force] [/hidefile][/delay=x] 
 
FichierHTM → chemin complet d’accès au fichier charte en HTML (un dossier comportant les images associées est 
aussi supporté, il doit s’appeler de la même manière que le fichier HTML suivi de « _fichiers » ou « _files ». Il s'agit 
en général d'un document Microsoft Word sauvé au format HTML. (Le nom du dossier associé est automatiquement 
détecté si celui-ci a été généré par Microsoft Word, sinon il sera supposé…). 
 
FichierReponse → chemin complet ou sera stocké un fichier texte qui indique notamment le fait que l’utilisateur ait 
accepté ou non la charte. Au lancement de KoXo Charter si ce fichier est trouvé et que le statut est « ACCEPT », alors 
KoXo Charter se ferme aussitôt. 
 
/force → Si le fichier de réponse (avec une acceptation) existe déjà on affiche quand même la charte. 
 
/hidefile → Indique qu’il faudra cacher le fichier de réponse 
 
/delay=x → Indique le délai en secondes pendant lequel on doit attendre avant de pouvoir accepter ou refuser. (La 
valeur par défaut étant de 60 secondes). 

Le script de lancement lit le dossier réseau personnel de l'utilisateur, il essaye de copier le binaire sur la 

racine du disque dur local afin de minimiser la charge réseau pour les utilisateurs suivants. Si l'exécutable change de 
date alors il est remplacé. L'année scolaire est ensuite déterminée, elle est utilisée pour nommer le fichier réponse.  
Par défaut la charte Html s'appelle « Charte.Htm » il doit se trouver (avec son éventuel dossier associé) dans le 
même dossier que le script VBS. Le fichier de réponse est écrit dans le sous dossier « CONFIG » du dossier réseau 
personnel de l'utilisateur. 
 
Le fichier VBS fourni ne cache pas le fichier de réponse, pour le faire, il suffit de modifier la quatrième ligne en 
partant de la fin: 

objShell.Run KoXoCharter & " " & CharterHTMFilePath & " " & StatusFilePath & " /hidefile", 1, True 
 
NB : Il faut que l’attribut «Active Directory»  "home" soit correctement renseigné au niveau de l’utilisateur dans 
l’annuaire. 
 
Script de lancement «KoXoCharter.vbs» :  
 
'/////////////////////////////////////////////////////////////////////////////////////////// 
'//                           (C)2006-2019 KoXo Développement                             // 
'//                                     KoXo Charter V2.0                                 // 
'//                        Display the IT charter for the logged user                     // 
'/////////////////////////////////////////////////////////////////////////////////////////// 
On Error Resume Next 
Dim objNetwork, objShell, objDomain, objConnection, objCommand, objRecordSet, objUser 
Dim StatusFilePath, CharterFileName, ScholarYear, Quote, CommandLine 
StatusFilePath="CONFIG\Charter" 
CharterFileName="Charte.htm" 
ScholarYear=True 
Quote=Chr(34) 
Set objNetwork=WScript.CreateObject("WScript.Network") 
Set objShell=WScript.CreateObject("WScript.Shell") 
strUserName=objNetwork.UserName 
Set objDomain=GetObject("LDAP://rootDse") 
objDC=objDomain.Get("defaultNamingContext") 
Set objConnection=CreateObject("ADODB.Connection") 
objConnection.Open "Provider=ADsDSOObject;" 
Set objCommand=CreateObject("ADODB.Command") 
objCommand.ActiveConnection=objConnection 
objCommand.CommandText="<LDAP://" & objDC & ">;(&(objectCategory=User)" & _ 
                       "(samAccountName=" & strUserName & "));distinguishedName;subtree" 
Set objRecordSet=objCommand.Execute 
If objRecordset.RecordCount=1 Then 
   objRecordSet.MoveFirst 
   LDAP="LDAP://"+objRecordSet.Fields("distinguishedName").Value 
   Set objUser=GetObject(LDAP) 
   homeDirectory=objUser.Get("homeDirectory") 
   'If homeDirectory exists 
   If homeDirectory<>"" Then 
      StatusFilePath=homeDirectory & "\" & StatusFilePath & _ 
                     GetCurrentYear(ScholarYear) & ".TXT" 
      'KoXo Charter is executed if necessary only 
      If Not CharterIsAccepted(StatusFilePath) Then 
         CharterHTMFilePath=Replace(WScript.ScriptFullName, _ 
                                    WScript.ScriptName, _  
                                    CharterFileName) 
         If OsIs64bits Then  
            KoXoCharter=Replace(WScript.ScriptFullName, _  
                                WScript.ScriptName, _  
                                "KoXoCharter64.exe") 
            Else 
            KoXoCharter=Replace(WScript.ScriptFullName, _  
                                WScript.ScriptName, _  

                                "KoXoCharter32.exe") 
            End If                     
         LocalKoXoCharter=CopyKoXoCharterOnLocalDrive(KoXoCharter) 
         If LocalKoXoCharter<>"" Then  
            KoXoCharter=LocalKoXoCharter 
            End If         
         err.Number=0 
         CommandLine=Quote & _ 
                     KoXoCharter & Quote & " " & Quote & _ 
                     CharterHTMFilePath & Quote  & " " &  Quote & _ 
                     StatusFilePath & Quote  
         objShell.Run CommandLine, 1, True    
         If err.number<>0 Then 
            objShell.Popup "Execution : " & CommandLine & VbCrLf & _ 
                           "Description=" & err.Description & VbCrLf & _  
                           "Code=" & err.Number,5,"Erreur",64 
            End If 
         End If 
      End If  
   End If    
objConnection.Close 
WScript.Quit 
'/////////////////////////////////////////////////////////////////////////////////////////// 
' Try to copy KoXoCharterXX.exe on local drive 
Function CopyKoXoCharterOnLocalDrive(KoXoCharterSourceFullPath) 
On Error Resume Next 
Dim WshShell, Fso 
Dim KoXoCharterLocalFolder, SrcFile, DstFile, KoXoCharterBinaryFullPath 
Set WshShell=WScript.CreateObject("WScript.Shell") 
Set Fso=CreateObject("Scripting.FileSystemObject") 
KoXoCharterLocalFolder=WshShell.ExpandEnvironmentStrings("%systemdrive%") & "\KoXoCharter" 
If Not(Fso.FolderExists(KoXoCharterLocalFolder)) Then  
   Fso.CreateFolder(KoXoCharterLocalFolder) 
   End If 
If OsIs64bits Then  
      KoXoCharterBinaryFullPath=KoXoCharterLocalFolder & "\KoXoCharter64.exe"  
      Else  
      KoXoCharterBinaryFullPath=KoXoCharterLocalFolder & "\KoXoCharter32.exe"  
      End If     
If Fso.FolderExists(KoXoCharterLocalFolder) Then  
    If Not(Fso.FileExists(KoXoCharterBinaryFullPath)) Then  
       Fso.CopyFile KoXoCharterSourceFullPath, KoXoCharterBinaryFullPath 
      End If 
    Set SrcFile=Fso.GetFile(KoXoCharterSourceFullPath) 
    Set DstFile=Fso.GetFile(KoXoCharterBinaryFullPath)  
    If (FileSource.DateLastModified<>FileDest.DateLastModified) Then 
       Fso.CopyFile KoXoCharterSourceFullPath, KoXoCharterBinaryFullPath 
       End If   
   End If 
If Fso.FileExists(KoXoCharterBinaryFullPath) Then    
   CopyKoXoCharterOnLocalDrive=KoXoCharterBinaryFullPath 
   Else 
   CopyKoXoCharterOnLocalDrive="" 
   End If 
End Function 
'/////////////////////////////////////////////////////////////////////////////////////////// 
' Is Charter accepted ? 
Function CharterIsAccepted(CharterStatusFile) 
Dim objFSO, Res, i, strLine, Status 
 Set objFSO=CreateObject("Scripting.FileSystemObject") 
 Res=False 
 If (objFSO.FileExists(CharterStatusFile)) Then   
    Set objTextFile=objFSO.OpenTextFile(CharterStatusFile,1) 
        Do Until objTextFile.AtEndOfStream 
           strLine=UCase(objTextFile.ReadLine) 
           i=InStr(strLine,"STATUS=")  
           If i>0 Then 
              Status=Mid(strLine,8,255) 



Utilisation de KoXo Charter V2.0 
http://www.koxo.net 

 

 Page 8/9 © 2006-2019 KoXo Développement   

              If InStr("ACCEPT",Status)>0 Then  
                 Res=True 
                 Exit Do 
                 End If 
              End If 
        Loop 
    objTextFile.Close 
    End If 
 CharterIsAccepted=Res 
End Function 
'/////////////////////////////////////////////////////////////////////////////////////////// 
' Get the current year // -> 2020 or 2019-2020 
Function GetCurrentYear(ScholarYear) 
Dim ScholarDay, ScholarMonth 
ScholarMonth=9  'September 
ScholarDay=1    '1st 
If ScholarYear Then 
   If DateDiff("s",DateSerial(Year(Date),ScholarMonth,ScholarDay),Now)>0 Then 
      CurrentYear=Year(Date) & "-" & Year(Date)+1 
      Else  
      CurrentYear=Year(Date)-1 & "-" & Year(Date) 
      End If     
   Else 
    CurrentYear=Year(Date) 
   End If     
 GetCurrentYear=CurrentYear   
End Function 
'/////////////////////////////////////////////////////////////////////////////////////////// 
'64 bits ? 
Function OsIs64bits 
On Error Resume Next 
Dim objWMIService,colOperatingSystems 
Set objWMIService=GetObject("winmgmts:" & "{impersonationLevel=impersonate}!\\.\root\cimv2") 
Set colOperatingSystems=objWMIService.ExecQuery("Select * from Win32_OperatingSystem") 
For Each objOperatingSystem in colOperatingSystems 
    OSArchi=objOperatingSystem.OSArchitecture 
Next 
OsIs64bits=InStr(OSArchi,"64") 
End Function 
'/////////////////////////////////////////////////////////////////////////////////////////// 

5 – Annexes 

 
Les variables utilisables dans la page HTML: 
 

Variable Description 
%TITLE% Titre, civilité (M., Mle, Mme, Mrs, Mr …) 

%FUNCTION% Fonction (Dr, Pr, …) 

%LAST_NAME% Nom de l’utilisateur 

%FIRST_NAME% Prénom de l’utilisateur 

%EMAIL% Email de l’utilisateur 

%FQDN% Nom complet de l’utilisateur dans l’annuaire 
Ex :CN=jdupont,OU=Magasin,OU=Usine,OU=Utilisateurs,OU=K
oXoAdm,DC=koxo,DC=net 

%HOME_DRIVE% Lettre de lecteur personnel de l’utilisateur 

%HOME_DIRECTORY% Chemin du répertoire de base de l’utilisateur 
Ex : \\serv\jdupont$ 

%PROFILE_PATH% Chemin du profil de l’utilisateur 
Ex : \\serv\jdupont$\profil 

%CONTAINER% Nom du conteneur Active Directory ou se trouve l’utilisateur 

%DEPARTMENT% Service ou département de l’utilisateur 

%COMPANY% Nom de la compagnie, société 

%COMPUTER_NAME% Nom de la station sur laquelle l’utilisateur est logué 

%NETBIOS_DOMAIN% Nom NetBIOS du domaine 

%DNS_DOMAIN% Nom DNS du domaine 

%LDAP_DOMAIN% Nom LDAP du domaine 

%DATE% Date au format court  
Ex : 12/08/2019 

%LONG_DATE% Date au format long 
Ex : lundi 12 aout mars 2019 

%TIME% Heure au format court 
17:43 

%LONG_TIME% Heure au format long 
18:23:20 

 
 


